
Note:  If the binary number is less than 10, then there is no need to enter the Double Dabble 
routine.  The binary number is the same as the BCD number.

This code is an implementation of the "unified pre-correction and shift loop" idea.  Instead of 
using the complicated method of counting the binary shifts and adding an extra shift if a pre-
correction is done... this code uses the idea of the "active BCD digit count" to control the 
iteration of the unified loop.

At the start of the routine, the active BCD digit count is 1.  Initially only one active BCD digit will 
be processed by the unified loop.... the low-order BCD digit.  Bits of the binary number are 
shifted in, the test for pre-correction is done, and then the BCD digit is shifted left.  

After each termination of the unified loop, a check is made to see if DF (carry) bit  is set to 1.  
(Eventually the carry bit will be set to 1 at the end of the unified loop.)  If the carry bit is set to 1 
here, then a new BCD digit is created  by storing a 1 in the next available BCD digit. 
Additionally, the active BCD digit count is incremented.   This is how each of the next" BCD 
digits are created and initialized. Now when the main loop shifts in a new binary digit, the 
unified loop will process another active BCD digit.

Note that at the end of the unified loop, the last BCD digit processed will be the current high-
order BCD digit.  (The BCD digits are processed in order from least-significant to most 
significant.) A 1 bit shifted into the carry from the high-order BCD digit indicates that a new 
high-order BCD digit should be created.  That BCD digit is created and initialized to 1... when 
the test-for-carry-bit-set is done after the unified loop terminates.  This paragraph is the 
justification for creating the new high-order BCD digit.

To make the unified pre-correction/shift loop work, the carry bit must be preserved before the 
pre-correction code is executed.  Otherwise the pre-correction test code will wipe out the value 
shifted into the carry bit.  After the pre-correction code is done, the carry value that was shifted 
in... will be restored to the DF (carry) register.

The low-order BCD digit begins with a value of 1. The loop that shifts out and discards the 
high-order non-significant zeros from the binary number... stops when the carry bit is set to 1.  
This value of 1 never has a chance to shift into the BCD digits, so we start the low-order BCD 
digit as 1 to "capture": that original 1 carry value.

Adding 120 to a BCD digit when it is pre-corrected... is still done so that the high-order bit of 
the BCD digit will become zero and the high-order bit of the byte will become 1 and shift into 
the carry bit.

The A18 assembler was used here and it does not support macros.

;
; Register Definitions:
;
R0               EQU      0
R1               EQU      1
R2               EQU      2
R3               EQU      3
R4               EQU      4
R5               EQU      5
R6               EQU      6
R7               EQU      7
R8               EQU      8
R9               EQU      9
R10             EQU     10
R11             EQU     11



R12             EQU     12
R13             EQU     13
R14             EQU     14
R15             EQU     15

; **********************************
; R6  - BINARY BIT COUNT
; R8 & R9 - BINARY NUMBER
; R12 - INDEX REGISTER
; R13 - LOW BYTE STORES LOCAL LOOP COUNTER
; R13 - HIGH BYTE SAVES CURRENT BCD DIGIT
; R14 - LOW BYTE STORES ACTIVE BCD DIGIT COUNT
; R14 - HIGH BYTE SAVES CARRY VALUE (DF)
; **********************************

This reserves the space for the BCD digits.

BGNDIG:          DB    0,0,0,0,0,0,0,0,0
ENDDIG:          DB    1

This is the test binary value to be converted to BCD.

CONVRT:         DB    $00,$CE,$ED,$BB         ; DECIMAL 847,291

START             ORG     100H

Set up to make Register 12 the index register.

; -----------------
; *** SET UP INDEX REGISTER LOAD
                        LDI          HIGH(ENDDIG)
                        PHI          R12
                        LDI          LOW(ENDDIG)
                        PLO        R5

Load the binary number to be converted into Registers 8 and 9.

; -----------------
; *** LOAD BINARY NUMBER TO CONVERT INTO R8 AND R9
                        LDI          HIGH(CONVRT)
                        PHI          R13
                       LDI           LOW(CONVRT)
                       PLO         R13
                       LDA         R13
                       PHI          R8
                       LDA         R13
                       PLO         R8
                       LDA         R13
                       PHI          R9
                       LDA          R13
                       PLO          R9

Store a constant 1 in the high byte of R6 and also in the initial active BCD digit count.

; -----------------
; *** SET UPPER BYTE OF R6 TO 1
                      LDI           1



                      PHI          R6              ;STORE CONSTANT 1 INTO HIGH BYTE OF R6
                      PLO         R14           ;ACTIVE BCD DIGIT COUNT -- STARTS AT 1

Set up count for shifting out high-order zeros from binary number.

; -----------------
; *** SET BINARY BIT COUNT IN LOW BYTE OF R6
                      LDI     32
                      PLO     R6

Loop to shift out high-order zero bits from binary number until carry = 1.

; -----------------
; SHIFT OUT HIGH ORDER ZERO BITS OF BINARY NUMBER
SHFOUT:      GLO     R9
                     SHLC
                     PLO     R9
;               -----------------
                     GHI     R9
                     SHLC 
                     PHI     R9
;               -----------------
                    GLO     R8
                    SHLC
                    PLO     R8
;               -----------------
                    GHI     R8
                    SHLC 
                    PHI     R8
;               -----------------
                    DEC     R6              ;DECREMENT BINARY BIT COUNT
                    BNF     SHFOUT

Here is the main Double Dabble code.

;
; -----------------
; *** MAIN DOUBLE DABBLE LOOP
NLOOP:  Beginning of the main Double Dabble loop.

Shift the next bit of the binary number into the DF (carry) bit.

; -----------------
; SHIFT OUT HIGH-ORDER BIT OF BINARY NUMBER
; -----------------
                    GLO     R9
                    SHLC
                    PLO     R9
;               -----------------
                    GHI     R9
                    SHLC 
                    PHI     R9
;               -----------------
                    GLO     R8
                    SHLC
                    PLO     R8
;               -----------------



                    GHI     R8
                    SHLC 
                    PHI     R8

Set the lower byte of the index register R12 to the lowest address of the BCD digit 
area--the upper byte is not changed by Double Dabble.  Also, the PHI, PLO, GHI, and 
GLO instructions do not affect the carry bit.  So the carry value shifted out from the 
binary number is still safe after setting the index register and loop counter values 
below.

; -----------------
; SET INDEX REGISTER (LOW BYTE ONLY)
                    GLO     R5              ;GET INDEX LOW BYTE
                    PLO     R12             ;STORE IN INDEX REG.

Copy the active BCD count from the lower byte of R14 to the lower byte of the R13 loop 
counter.

; -----------------
; CALCULATE LOOP COUNT
                    GLO     R14             ;GET ACTIVE BCD DIGIT COUNT
                    PLO     R13             ;STORE LOCAL LOOP COUNT

UNILP:    Beginning the unified pre-correction and shift loop.

Save the current value of DF (carry) by shifting the carry bit into the low order bit of the 
D register and storing the D register in the upper byte of R14.

; -----------------
; *** SAVE DF (CARRY) IN UPPER BYTE OF R14
                    SHLC                     ;GET CARRY TO LOW-ORDER BIT OF D
                    PHI     R14             ;SAVE D IN UPPER BYTE OF R14

Test for and do the pre-correction if necessary.  The active BCD digit is loaded into the 
D register from memory using the index register R12. The BCD digit is immediately 
stored in the upper byte of R13.  An immediate 5 is subtracted from D.  If the BCD digit 
is five or greater, a pre-correction is done.  The 3 pre-correction is added, plus the 
restoration of the 5 subtracted, plus 120... which is done by adding an immediate 128.

The BCD digit is then stored in R13.  Whether the pre-correction is done or skipped... 
the value of the current BCD digit will be found in the upper byte of R13 by the shift 
code.

; ----------------
; *** TEST ONE BYTE FOR PRE-CORRECTION                
                    LDX                       ;LOAD D FROM INX ADDRESS
                    PHI     R13             ;SAVE BCD DIGIT IN UPPER BYTE OF R13
                    SMI     5                 ;IS D > 4
                    BNF     NOFIX        ;SKIP CORRECTION IF D < 0
; *** PRE-CORRECT BCD DIGIT
                    ADI     128             ;ADD 3 CORRECTION + 5 RESTORE + 120
                    PHI     R13             ;SAVE BCD DIGIT IN UPPER BYTE OF R13
NOFIX:
; -----------------
; SHIFT ONE BCD DIGIT
; -----------------



Restore the carry value that was previously shifted out into carry. (Get the stored carry 
value from the upper byte of R14 and shift the low-order bit into the DF (carry) register.

                    GHI     R14             ;GET CARRY BACK TO D
                    SHR                       ;PUT CARRY VALUE INTO DF

Get the current BCD digit from the upper byte of R13.

                    GHI     R13             ;GET WORKING BCD DIGIT

Shift the carry into the lower bit of the D register and shift out the high-oder D register 
bit into carry.  Store the resulting BCD digit back into memory using index register R12.  
Note that the STXD instruction also *decrements* the index register after storing the 
value.  Now we are ready to process the next BCD digit.

                    SHLC                     ;ROTATE D THROUGH CARRY
                    STXD                     ;STORE D AND DECREMENT INDEX REGISTER
; -----------------
; *** UNILP COUNTER OVERHEAD

This code decrements the loop count and repeats the unified loop (UNILP) if the count 
is *not* zero.  None of these instructions will affect the value in the DF (carry) register. 
Also note that the first thing done in the unified loop is to save the value of the 
shifted-out carry bit.

                    DEC     R13             ;DECREMENT CORRECT-SIFT LOOP COUNTER
                    GLO     R13             ;D = LOCAL LOOP COUNT
                    BNZ     UNILP          ;LOOP UNTIL DONE

; -----------------
; *** IF DF = 1, CREATE NEW BCD DIGIT = 1 -- INCREMENT ACTIVE BCD DIGIT COUNT

This code is executed after the termination of the unified loop (UNILP) code.  If the 
loop leaves the carry bit set to 1, then a *new* active BCD digit is created by setting the 
next BCD digit to 1 (as if the carry had been shifted into that BCD digit).  The active 
BCD digit count is incremented to show that there is one more active BCD digit to be 
processed by UNILP.  (Note that after the unified loop (UNILP) terminates, the index 
register will be left pointing to the next unused, inactive BCD digit and that digit will 
become the next active BCD digit.

                    BNF      SKPDIG          ;IF CARRY IS ZERO, NO NEW BCD DIGIT
                    INC       R14                ;INCREMENT ACTIVE BCD DIGIT COUNT
                    GHI       R6                 ;GET A BYTE WITH VALUE OF 1
                    STXD                          ;STORE D IN NEW BCD DIGIT POSITION

; -----------------
; *** MAIN LOOP COUNTER OVERHEAD

This is the test code for exiting the main loop.  The low-order byte of register R6 
contains the number of bits of binary number to shift into the BCD digit area.  When the 
count is decremented to zero, Double Dabble has processed the entire binary number 
and is done.

SKPDIG:      DEC     R6                 ;DECREMENT BINARY BIT COUNT
                    GLO     R6                 ;D = BINARY BIT COUNT
                    BNZ     NLOOP          ;LOOP UNTIL DONE



                    IDL
                    END


